M7. Ábrák programkódjai

2.1 A fékezetlen inga mozgása

2.2 A lineáris matematikai inga

2.1 ábra

```
Clear[k, var, matinga, x1, x2, y1, y2, t0, t1, Vmatinga, ContourSwingVmat,
  SwingFieldmat, IC, PendulumTrajmat, PendPlotxmat, PendPlotmat];
k = 1;
var = {x, y};
matinga := {y, -k^2 x};
x1 = -π; x2 = π; y1 = -3; y2 = 3;
t0 = 0; t1 = 4 Pi;
Vmatinga[{x_, y_}] = y^2/2 + k^2 x^2/2;
```

```
ContourSwingVmat = ContourPlot[Vmatinga[var], {x, x1, x2},
   {y, y1, y2}, ContourShading → False, ImageSize → 120, AxesLabel → {x, y}];
SwingFieldmat = VectorFieldPlot[matinga, {x, x1, x2}, {y, y1, y2}, Axes → True,
   ScaleFactor → 1, PlotRange → {-π, π}, ImageSize → 120, AxesLabel → {x, y}];
IC = Table[ N[{0., u}], {u, 1.4, 3, 0.2}];
PendulumTrajmat[t_] = ODESolve[matinga, var, IC, {t, t0, t1}];
PendPlotxmat =
  Plot[PendulumTrajmat[t][[All, 1]], {t, t0, t1}, ImageSize → 120, AxesLabel → {t, x}];
PendPlotmat = ParametricPlot[PendulumTrajmat[t], {t, t0, t1},
   PlotRange → {-π, π}, ColorFunction → (Hue[#3 / (2 Pi)] &),
   ColorFunctionScaling → False, ImageSize → 120, AxesLabel → {x, y}];
Row[{PendPlotxmat, Show[SwingFieldmat, ContourSwingVmat, PendPlotmat]}]
```

2.3 A nemlineáris matematikai inga

2.2 ábra

A 2.2 ábra programkódja a 2.1 ábra programkódjához hasonló, annyi eltéréssel, hogy itt

```
fizinga := {y, -k^2 Sin[x]};
x1 = -π; x2 = 4.9 π; y1 = -3; y2 = 3;
t0 = 0; t1 = 10 π;
Vfiz[{x_, y_}] = y^2/2 + ∫_0^x Sin[u] ⅆu;
IC = Table[ N[{0., u}], {u, 1.4, 2.8, 0.2}];
```

illetve a harmadik koordináta szetint színezünk, és nem normáljuk le 2π-vel.

Interaktív illusztráció

```
Manipulate[
 DynamicModule[{eqns1, eqns2, sol1, sol2, x, y, t, s},
  eqns1 = {x'[t] == y[t], y'[t] == -rug x[t], x[0] == w1, y[0] == w2};
  eqns2 = {x'[t] == y[t], y'[t] == -rug Sin[x[t]], x[0] == w1, y[0] == w2};
  sol1[s_] = {x[s], y[s]} /.NDSolve[eqns1, {x, y}, {t, 0, 60}];
  sol2[s_] = {x[s], y[s]} /.NDSolve[eqns2, {x, y}, {t, 0, 60}];
  Column[{Row[{
      Graphics[{Point[{0, 0}], Line[{{0, 0}, {Sin[sol1[p][[1, 1]]], -Cos[sol1[p][[1, 1]]]}}],
        Darker[Red], Disk[{Sin[sol1[p][[1, 1]]], -Cos[sol1[p][[1, 1]]]}, .05]},
       PlotRange → {{-1.2, 1.2}, {-1.2, 1.2}},
```

```
      PlotLabel → Style["Lineáris  inga", Tiny], ImageSize → {140, 140}],
     Graphics[{Point[{0, 0}], Line[{{0, 0}, {Sin[sol2[p][[1, 1]]],
          -Cos[sol2[p][[1, 1]]]}}], Darker[Brown, 0.5],
       Disk[{Sin[sol2[p][[1, 1]]], -Cos[sol2[p][[1, 1]]]}, .05]},
      PlotRange → {{-1.2, 1.2}, {-1.2, 1.2}},
      PlotLabel → Style["Nemlineáris inga", Tiny], ImageSize → {140, 140}]
    }],
   Row[{
     Show[Plot[Evaluate[sol1[t][[1, 1]]], {t, 0, p}, AxesOrigin → {0, 0},
       AxesLabel → {"t", "x"}, PlotRange → {{0, 60}, {-r, r}}, PlotStyle → Red,
       ImageSize → {140, 140}], Plot[Evaluate[sol2[t][[1, 1]]], {t, 0, p},
       AxesOrigin → {0, 0}, AxesLabel → {"t", "x"}, PlotRange → {{0, 60}, {-r, r}},
       PlotStyle → Darker[Brown, 0.3], ImageSize → {140, 140}],
      Graphics[{Darker[Brown], PointSize → 0.05, Point[{p, sol2[p][[1, 1]]}]}]
      , Graphics[{Darker[Red], PointSize → 0.05, Point[{p, sol1[p][[1, 1]]}]}]],
     Show[ParametricPlot[Evaluate[sol1[t][[1]]], {t, 0, p}, AxesOrigin → {0, 0},
       AxesLabel → {"x", "y"}, PlotStyle → Red, ImageSize → {140, 140}], Graphics[
       {Darker[Red], PointSize → 0.05, Point[{sol1[p][[1, 1]], sol1[p][[1, 2]]}]}],
      ParametricPlot[Evaluate[sol2[t][[1]]], {t, 0, p}, AxesOrigin → {0, 0}, AxesLabel →
        {"x", "y"}, PlotStyle → Darker[Brown, 0.3], ImageSize → {140, 140}], Graphics[
       {Darker[Brown], PointSize → 0.05, Point[{sol2[p][[1, 1]], sol2[p][[1, 2]]}]}],
      ContourPlot[y^2 / 2 + rug (1 - Cos[x]), {x, -π, r}, {y, -π, π},
       ContourShading → False, Contours → {2 rug},
       ContourStyle → {{Blue, Opacity[0.2]}}], PlotRange → {{-π, r}, {-π, π}}]
    }]}, Frame → All]
 ],
 {{rug, 1, "Rugalmassági együttható (k^2)"},
  0, 5, Appearance → "Labeled", ImageSize → Tiny},
 {{w1, 2, "Kezdeti kitérés"}, -π, π, Appearance → "Labeled", ImageSize → Tiny},
 {{w2, 2, "Kezdeti sebesség"}, 0, 2, Appearance → "Labeled", ImageSize → Tiny},
 {{r, 7, "Kicsinyítés"}, 2, 20, Appearance → "Labeled", ImageSize → Tiny},
 {{p, 0.01, "Idő"}, 0.01, 60, ControlType → Manipulator,
  Appearance → "Open", AnimationRate → 1, ImageSize → Tiny},
 AutorunSequencing → {8}, SaveDefinitions → True, ControlPlacement → Bottom]
```

3. Fékezett ingamozgás

3.1 A fékezett lineáris matematikai inga

3.1 ábra

A 3.1 ábra programkódja a 2.1 ábra programkódjához hasonló, annyi eltéréssel, hogy itt

```
a = 0.2;
matcsinga := {y, -a y - k^2 x};
t0 = 0; t1 = 20.;
```

illetve a harmadik koordináta szetint színezünk, és nem normáljuk le 2π-vel.

3.2 A fékezett nemlineáris matematikai inga

3.2 ábra

A 3.2 ábra programkódja szintén a 2.1 ábra programkódjához hasonló, annyi eltéréssel, hogy itt

```
a = 0.2;
fizingcs := {y, -a y - Sin[x]};
x1 = -π; x2 = 3 π; y1 = -3; y2 = 3;
t0 = 0; t1 = 20.;
```

```
Vfiz[{x_, y_}] = y^2/2 + ∫_0^x Sin[u] ⅆu;
```

illetve a harmadik koordináta szetint színezünk, és nem normáljuk le 2π-vel.

Interaktív illusztráció

Az interaktív illusztráció programkódja a 2.3 részben található interaktív illusztráció programkódjához hasonló, annyi eltéréssel, hogy itt egyenleteink:

```
eqns1 = {x'[t] == y[t], y'[t] == -a y[t] - rug x[t], x[0] == w1, y[0] == w2};
eqns2 = {x'[t] == y[t], y'[t] == -a y[t] - rug Sin[x[t]], x[0] == w1, y[0] == w2};
```

4. Ingamozgás periodikus külső erő hatására

4.1 Fékezetlen ingamozgás periodikus külső erő hatására

4.1.1 Fékezetlen lineáris matematikai inga

4.1 ábra

```
A = 1; k = 1; ν = 2;
Column[
 {Table[Plot[Evaluate[{x[t], y[t]} /.NDSolve[{x'[t] == y[t], y'[t] == -k^2 x[t] + A Cos[ν t],
        x[0] == 0, y[0] == a}, {x[t], y[t]}, {t, 0, 60}][[1, 1]]],
    {t, 0, 60}, PlotRange → {-1.5 a, 1.5 a}, AxesLabel → { t , x}, PlotLabel →
     StringJoin["x(0)=0 ", "y(0)=", ToString[a]], ImageSize → 120], {a, 3, 6, 3}],
  Table[ParametricPlot[Evaluate[{x[t], y[t]} /.NDSolve[{x'[t] == y[t],
       y'[t] == -x[t] + A Cos[ν t], x[0] == 0, y[0] == a}, {x[t], y[t]}, {t, 0, 60}]],
    {t, 0, 60}, PlotRange → {-1.5 a, 1.5 a}, AxesLabel → { x , y},
    ImageSize → 120], {a, 3, 6, 3}]}]
```

4.2 ábra

```
Table[
 Column[
  {Row[{Show[Plot[(A (-Cos[k t]))/(k^2 - ν^2) /. {A → 1, k → 1}, {t, 0, 30}, PlotStyle → Red,
       ImageSize → 90, AxesLabel → {t, x}, PlotLabel → StringJoin["ν=", ToString[ν]]],
      Plot[(A (Cos[t ν]))/(k^2 - ν^2) /. {A → 1, k → 1}, {t, 0, 30},
       PlotStyle → Darker[Brown, 0.5], ImageSize → 90, AxesLabel → {t, x}]]}],
   Plot[(A (-Cos[k t] + Cos[t ν]))/(k^2 - ν^2) /. {A → 1, k → 1}, {t, 0, 30},
    ImageSize → 90, AxesLabel → {t, x}]}], {ν, 0.65, 1.19, 0.27}]
```

4.3 ábra

```
Plot[(A t Sin[k t])/(2 k) /. {A → 1, k → 1}, {t, 0, 100},
 AxesLabel → {t, x}, ImageSize → 120, PlotLabel → "k=ν=1, A=1"]
```

Interaktív illusztráció

```
Manipulate[Column[{
   Show[Plot[(A (-Cos[k t]))/(k^2 - ν^2), {t, 0, T}, PlotStyle → Red, ImageSize → 200, AxesLabel → {t, x}],
```

```
   Plot[A (Cos[t ν])/(k² - ν²), {t, 0, T}, PlotStyle → Darker[Brown],
     ImageSize → 160, AxesLabel → {t, x}]],
   Plot[A (-Cos[k t] + Cos[t ν])/(k² - ν²), {t, 0, T}, ImageSize → 200, AxesLabel → {t, x}]}],
 {{A, 1.2, "Amplitúdó"}, 1, 2, ImageSize → Tiny, Appearance → "Labeled"},
 {{k, 4000, "Természetes körfrekvencia"},
  1000, 8000, ImageSize → Tiny, Appearance → "Labeled"},
 {{ν, 4002, "Külső hatás körfrekvenciája"}, 1000, 8000,
  ImageSize → Tiny, Appearance → "Labeled"},
 {{T, 6.5, "Idő"}, 1, 10, ImageSize → Tiny, Appearance → "Labeled"},
 Button["Lejátszás", EmitSound[Sound[Play[{A (-Cos[k t])/(k² - ν²) + A (Cos[t ν])/(k² - ν²)}, {t, 0, T}]]],
  ImageSize → Tiny], ControlPlacement → Bottom]
```

Interaktív illusztráció

```
Manipulate[Row[{Plot[Evaluate[{x[t], y[t]} /.
       NDSolve[{x'[t] == y[t], y'[t] == -rug x[t] + A Cos[ν t], x[0] == w1, y[0] == w2},
         {x[t], y[t]}, {t, 0, T}, MaxSteps → 100000, MaxStepSize → 0.05][[1, 1]]],
     {t, 0, T}, PlotRange → {-r, r}, AxesLabel → { t , x}, ImageSize → 120],
   ParametricPlot[Evaluate[{x[t], y[t]} /.
       NDSolve[{x'[t] == y[t], y'[t] == -rug x[t] + A Cos[ν t], x[0] == w1, y[0] == w2},
        {x[t], y[t]}, {t, 0, T}, MaxSteps → 100000, MaxStepSize → 0.05]], {t, 0, T},
    PlotRange → {-r, r}, AxesLabel → { x , y}, ImageSize → 120]}, Frame → True],
 {{ν, 1, "Külső hatás körfrekvenciája"}, 0.0, 10., 0.05,
  Appearance → "Labeled", ImageSize → Tiny},
 {{A, 1, "Külső hatás amplitúdója"}, 0, 5, Appearance → "Labeled", ImageSize → Tiny},
 {{rug, 1, "Rugalmassági együttható (k²)"},
  0.1, 2, Appearance → "Labeled", ImageSize → Tiny},
 {{w1, 0, "Kezdeti kitérés"}, -π, π, Appearance → "Labeled", ImageSize → Tiny},
 {{w2, 1, "Kezdeti sebesség"}, 0, 2, Appearance → "Labeled", ImageSize → Tiny},
 {{r, 5, "Kicsinyítés"}, 5, 100, Appearance → "Labeled", ImageSize → Tiny},
 {{T, 5, "Idő"}, 5, 200, Appearance → "Labeled", ImageSize → Tiny},
 ControlPlacement → Bottom]
```

4.1.2 Fékezetlen nemlineáris matematikai inga

Interaktív illusztráció

Az interaktív illusztráció probgramkódja az 4.1.2 rész utolsó interaktív illusztrációjának programkódjához hasonló, annyi eltéréssel, hogy itt egyenletünk alakja:

```
{x'[t] == y[t], y'[t] == -k Sin [x[t]] + A Cos[ν t], x[0] == w1, y[0] == w2}
```

4.4 ábra

```
Table[
 Plot[Evaluate[{x[t], y[t]} /.NDSolve[{x'[t] == y[t], y'[t] == -Sin[x[t]] + Cos[t], x[0] == w,
       y[0] == 1}, {x[t], y[t]}, {t, 0, 60}][[1, 1]]], {t, 0, 60}, PlotRange → {-10, 40},
  AxesLabel → { t , x}, PlotLabel → StringJoin["x(0)=", ToString[w]],
  ImageSize → 100], {w, 0, 6, 3}]
```

4.5 ábra

```
Row[{Plot[Evaluate[{x[t], y[t]} /.
     NDSolve[{x'[t] == y[t], y'[t] == - Sin [x[t]] + Cos[1.32 t], x[0] == 0, y[0] == 1},
       {x[t], y[t]}, {t, 0, 500}][[1, 1]]], {t, 0, 500}, PlotRange → {-60, 20},
   AxesLabel → { t , x}, PlotLabel → StringJoin["ν=1.33"], ImageSize → 120],
  Plot[Evaluate[{x[t], y[t]} /.NDSolve[{x'[t] == y[t], y'[t] == - Sin [x[t]] + Cos[1.35 t],
        x[0] == 0, y[0] == 1}, {x[t], y[t]}, {t, 0, 500}][[1, 1]]], {t, 0, 500},
   PlotRange → {-5, 5}, AxesLabel → { t , x}, PlotLabel → StringJoin["ν=1.35"],
```

```
    ImageSize → 120]}]
```

4.6 ábra

```
Clear[A, ν];
ν = 2;
Table[Plot[
  Evaluate[{x[t], y[t]} /.NDSolve[{x'[t] == y[t], y'[t] == -Sin[x[t]] + A Cos[ν t], x[0] == 0,
       y[0] == 0}, {x[t], y[t]}, {t, 0, 60}][[1, 1]]], {t, 0, 60},
  PlotRange → {-3, 12}, AxesLabel → { t , x}, PlotLabel → StringJoin["A=", ToString[A]],
  ImageSize → 100], {A, 3.07, 3.13, 0.03}]
```

4.2 Fékezett ingamozgás periodikus külső erő hatására

4.2.1 Fékezett lineáris matematikai inga

4.7 ábra

```
Clear[a, k, ν, A];
a = 1; k = 1; ν = 2; A = 1;
Column[{Table[
   Plot[Evaluate[{x[t], y[t]} /.NDSolve[{x'[t] == y[t], y'[t] == -a y[t] - k^2 x[t] + A Cos[ν t],
         x[0] == 0, y[0] == b}, {x[t], y[t]}, {t, 0, 60}][[1, 1]]],
    {t, 0, 60}, PlotRange → {-b / 2, b}, AxesLabel → { t , x}, ImageSize → 120,
    PlotLabel → StringJoin["x(0)=0 ", "y(0)=", ToString[b]]], {b, 3, 6, 3}],
  Table[ParametricPlot[Evaluate[{x[t], y[t]} /.NDSolve[{x'[t] == y[t],
        y'[t] == -x[t] + A Cos[ν t], x[0] == 0, y[0] == b}, {x[t], y[t]}, {t, 0, 60}]],
    {t, 0, 60}, PlotRange → {-1.2 b, 1.2 b}, AxesLabel → { x , y},
    PlotLabel → {0, b}, ImageSize → 120], {b, 3, 6, 3}]}]
```

4.8 ábra

```
Table[Plot[
  Evaluate[{x[t], y[t]} /.NDSolve[{x'[t] == y[t], y'[t] == -0.2 y[t] - x[t] + A Cos[t], x[0] == 0,
       y[0] == 0}, {x[t], y[t]}, {t, 0, 60}][[1, 1]]], {t, 0, 60},
  PlotRange → {-10, 10}, AxesLabel → { t , x}, PlotLabel → StringJoin["A=", ToString[A]],
  ImageSize → 100], {A, 1, 2, 0.5}]
```

Interaktív illusztráció

Az interaktív illusztráció programkódja a 4.1.2 rész utolsó interaktív illusztrációjának programkódjához hasonló, annyi eltéréssel, hogy itt egyenletünk alakja:

```
{x'[t] == y[t], y'[t] == -a y[t] - k x[t] + A Cos[ν t], x[0] == w1, y[0] == w2}
```

4.2.2 Fékezett nemlineáris matematikai inga

Interaktív illusztráció

Az interaktív illusztráció probgramkódja a 2.3-as rész interaktív illusztrációjának programkódjához hasonló, annyi eltéréssel, hogy itt egyenleteink:

```
eqns1 = {x'[t] == y[t], y'[t] == -a y[t] - rug x[t] + A Cos[ν t], x[0] == w1, y[0] == w2};
eqns2 = {x'[t] == y[t], y'[t] == -a y[t] - rug Sin[x[t]] + A Cos[ν t], x[0] == w1, y[0] == w2};
```

5. Ingamozgás periodikus hosszváltozás

hatására

5.2 A mozgás vizsgálata

5.1 ábra

```
Row[{Plot[2 + 1 Sign[Cos[2 Pi t]], {t, 0, 2.5}, ImageSize → 100, AxesLabel → {t, x}],
  Plot[2 + Cos[2 Pi t], {t, 0, 2.5}, AxesLabel → {t, x}, ImageSize → 100],
  Plot[2 + 1 Sin[2 Pi t], {t, 0, 2.5}, AxesLabel → {t, x}, ImageSize → 100]}]
```

Interaktív illusztráció

```
Manipulate[DynamicModule[{eqns1, eqns2, sol1, sol2, x, y, t, s, l, p, l1, l2, l3},
  l1[t_] = h0 + ϵ Sign[Cos[2 Pi / T (t - τ)]] ;
  l2[t_] = h0 + ϵ Cos[2 Pi / T (t - τ)] ;
  l3[t_] = h0 + ϵ Sin[2 Pi / T (t - τ)] ;
  l[t_] = Which[ll === 1, l1[t], ll === 2, l2[t], True, l3[t]];
  eqns1 = {x′[t] == y[t]/l[t]^2, y′[t] == -l[t] x[t] - a y[t], x[0] == w1, y[0] == w2};
  eqns2 = {x′[t] == y[t]/l[t]^2,
    y′[t] == -l[t] Sin[x[t]] - a y[t], x[0] == w1, y[0] == w2};
  sol1[s_] = {x[s], y[s]} /.NDSolve[eqns1, {x, y}, {t, 0, TT}, MaxSteps → 20 000];
  sol2[s_] = {x[s], y[s]} /.NDSolve[eqns2, {x, y}, {t, 0, TT}, MaxSteps → 20 000];
  Animate[
   Row[{
     Column[{Graphics[{Point[{0, 0}],
         Line[{{0, 0}, {l[p] Sin[sol1[p][[1, 1]]], -l[p] Cos[sol1[p][[1, 1]]]}}],
         Darker[Red], Disk[{l[p] Sin[sol1[p][[1, 1]]], -l[p] Cos[sol1[p][[1, 1]]]}, .05]},
        PlotRange → {{-(h0 + ϵ + 0.1), h0 + ϵ + 0.1}, {-(h0 + ϵ + 0.1), h0 + ϵ + 0.1}},
        PlotLabel → Style["Lineáris  inga", Tiny], ImageSize → {120, 120}],
       Plot[Evaluate[{l[t], sol1[t]〚1, 1〛}], {t, 0, p + 0.1}, AxesOrigin → {0, 0},
        AxesLabel → {"t", "x"}, PlotRange → {{0, TT}, {-r, r}},
        PlotStyle → {Red, Blue, Black}, ImageSize → {120, 120}],
       Show[ParametricPlot[Evaluate[sol1[t]〚1〛], {t, 0, p + 0.1}, AxesOrigin → {0, 0},
         AxesLabel → {"x", "y"}, PlotStyle → Red, ImageSize → {120, 120}],
        Graphics[{Darker[Red], PointSize → 0.05, Point[{sol1[p]〚1, 1〛, sol1[p]〚1, 2〛}]}],
        PlotRange → {{-r, r}, {-r, r}}]}, Center],
     Column[{
       Graphics[{Point[{0, 0}], Line[{{0, 0},
           {l[p] Sin[sol2[p][[1, 1]]], -l[p] Cos[sol2[p][[1, 1]]]}}], Darker[Brown, 0.5],
         Disk[{l[p] Sin[sol2[p][[1, 1]]], -l[p] Cos[sol2[p][[1, 1]]]}, .05]},
        PlotRange → {{-(h0 + ϵ + 0.1), h0 + ϵ + 0.1}, {-(h0 + ϵ + 0.1), h0 + ϵ + 0.1}},
        PlotLabel → Style["Nemlineáris inga inga", Tiny], ImageSize → {120, 120}],
       Plot[Evaluate[{l[t], sol2[t]〚1, 1〛}], {t, 0, p + 0.1}, AxesOrigin → {0, 0},
        AxesLabel → {"t", "x"}, PlotRange → {{0, TT}, {-r, r}},
        PlotStyle → {Red, Blue, Black}, ImageSize → {120, 120}],
       Show[ParametricPlot[Evaluate[sol2[t]〚1〛], {t, 0, p + 0.1}, AxesOrigin → {0, 0},
         AxesLabel → {"x", "y"}, PlotStyle → Brown, ImageSize → {120, 120}],
        Graphics[{Darker[Brown], PointSize → 0.05, Point[{sol2[p]〚1, 1〛, sol2[p]〚1, 2〛}]}],
        PlotRange → {{-r, r}, {-r, r}}]}, Center]}],
   {{p, TT - 0.1, "t"}, 0, TT - 0.1, ImageSize → Tiny}, AnimationRunning → False,
   DefaultDuration → TT, DisplayAllSteps → False]
 ],
 {{ll, 1, "l(t)"}, {1 → "Lépcsős", 2 → "Cos", 3 → "Sin"}},
 {{a, 0, "a"}, 0, 1, Appearance → "Labeled",
  ImageSize → Tiny}, {{h0, 1, "h0"},
  0.1, 2, Appearance → "Labeled", ImageSize → Tiny},
 {{ϵ, 0, "ϵ"}, 0, h0, Appearance → "Labeled", ImageSize → Tiny},
 {{T, 2, "T"}, 0.1, 5, Appearance → "Labeled", ImageSize → Tiny},
 {{τ, 0, "τ"}, 0, 5, Appearance → "Labeled", ImageSize → Tiny},
 {{w1, 0, "x(0)"}, -π, π, Appearance → "Labeled", ImageSize → Tiny},
 {{w2, 1, "y(0)"}, 0, 2, Appearance → "Labeled", ImageSize → Tiny},
 {{r, 2, "Kicsinyítés"}, 0.5, 15, Appearance → "Labeled", ImageSize → Tiny},
 {{TT, 5, "Idő"}, 2, 100, ImageSize → Tiny, Appearance → "Labeled"},
```

```
AutorunSequencing → {8}, ControlPlacement → Bottom]
```